
Options for Data for SAS 

Papers
• Use real data from your studies

– You definitely need approval from your PIs

– You probably need IRB approval

• Use data provided by SAS in the SASHELP library

– Everyone has access to it

– It may not fit your goals and examples well

• Simulate your own data using SAS!

– You can include the SAS programming to create your 

simulated data in your SAS paper

– I will focus on simulating data from a discrete distribution



Option 1
Macro with RAND(“uniform”)



Creating Integers for Categorical Variables

Option 1—use a simple macro

• From Rick Wicklin’s October 5, 2015 blog post…
/* SAS macro that duplicates the Excel RANDBETWEEN 
function */

%macro RandBetween(min, max);

(&min + floor((1+&max-&min)*rand("uniform")))

%mend;

• The FLOOR function
– “Returns the largest integer that is less than or equal to the 

argument, fuzzed to avoid unexpected floating point 
results.”



Creating Integers for Categorical Variables

Option 1—use a simple macro (continued)

• From Rick Wicklin’s October 5, 2015 blog post…

data work.RandInt;

do i = 1 to 100;

x = %RandBetween(1, 10);

output;

end;

run;

• This will give you a data set 100 records of a variable, 

x, that consists of integers between 1 and 10



Creating Integers for Categorical Variables

Option 1—use a simple macro    (continued)

• The RAND function

– “Generates random numbers from a distribution that 

you specify”, in this case the Uniform distribution with 

default parameters of a = 1 and b = 0.  

– He could have used other distributions supported by 

the RAND function, like the Normal, rand(“normal”), 

or Lognormal, rand(“lognormal”).  



Creating Integers for Categorical Variables

Option 1—use a simple macro (continued)

• The RAND function

– In Rick Wicklin’s July 10, 2013 blog post, he 

recommends against using the older random number 

generating functions like RANUNI and RANNOR for 

several reasons.  

– Wicklin recommends using the more modern RAND 

function which was introduced in SAS 9.1 (Sarle and 

Wicklin, 2018) because the RAND function has better 

statistical properties due to better underlying 

algorithms.   



Option 2
RANTBL function



Creating Integers for Categorical Variables

Option 2—RANTBL

• The RANTBL function

– “Returns a random variate from a tabled probability 

distribution.”

– RANTBL(seed, p1, p2, …, pi, …, pn)

• seed “is a numeric constant, variable, or 

expression with an integer value. If seed ≤ 0, the 

time of day is used to initialize the seed stream.”

• pi “is a numeric constant, variable, or expression” 

between 0 and 1.



• The RANTBL function
– “The RANTBL function returns a variate that is generated from 

the probability mass function defined by p1 through pn.  An 

inverse transform method applied to a RANUNI uniform variate 

is used. RANTBL returns…

• 1 with the probability p1

• 2 with the probability p2

⋮
• n with the probability pn

Creating Integers for Categorical Variables

Option 2—RANTBL     (continued)



• The RANTBL function
data work.RandomTable4EvenCats;

do i = 1 to 900;

r_4categories=rantbl(123, 0.25, 0.25, 0.25, 0.25) ; 

output;

end;

run;

– This will give you a data set 900 records of a variable, 

r_4categories, that consists of integers between 1 and 4

Creating Integers for Categorical Variables

Option 2—RANTBL (continued)



• The RANTBL function
data work.RandomTable5EvenCats;

do i = 1 to 900;

r_5equalcategories=rantbl(123, 0.2, 0.2, 0.2, 0.2, 0.2) ; 

output;

end;

run;

– This will give you a data set 900 records of a variable, 

r_5equalcategories, that consists of integers between 

1 and 5

Creating Integers for Categorical Variables

Option 2—RANTBL     (continued)



• One caveat with the RANTBL function…
– “The RANTBL function returns a variate that is generated from 

the probability mass function defined by p1 through pn.  An 

inverse transform method applied to a RANUNI uniform variate 

is used. RANTBL returns…

• Note that RANTBL uses the RANUNI function, 

which may be fine for many presentations and 

papers, but is likely insufficient for more 

advanced statistical and simulation work

Creating Integers for Categorical Variables

Option 2—RANTBL     (continued)



Option 3
The “Table” distribution with 

the RAND function



• The RAND(“Table”) function
data work.RandomTable5UnevenCats;

call streaminit (123) ; 

p1 = 0.1 ; p2 = 0.3 ; p3 = 0.1 ; p4 = 0.25 ; p5 = 0.25 ; 

do i = 1 to 900;

r_5unequal = rand("Table", p1, p2, p3, p4, p5); 

output;

end;

run;

– This will give you a data set 900 records of a variable, 

r_5unequal, that consists of integers between 1 and 5

– call streaminit is a call routine used with random number 

generators to specify the seed value

Creating Integers for Categorical Variables

Option 3—RAND(“Table”)



• The RAND(“Table”) function
– In Rick Wiklin’s Simulating Data with SAS (2013, page 19), he 

shows an additional example that uses a temporary array for the 

p values (this example uses 4 categories):

array  p[4]  _temporary_  (0.2, 0.2, 0.3, 0.3)  ;  

do i = 1 to 900 ; 

r_4unequal = rand( "Table", of p[*] ) ; 

output;

end;

– See Chapter 2, section 2.4.5, for additional information

Creating Integers for Categorical Variables

Option 3—RAND(“Table”) (continued)



Option 4
New in SAS 9.4M4:  The “Integer” 

distribution with the RAND function



• The RAND(“Integer”) function
– In Rick Wicklin’s October 5, 2015 blog post, he shows an 

additional example that uses the new “Integer” distribution that 

supports min and max parameters:

data RandInt;

do i = 1 to 100 ; 

x = rand("Integer", 1, 10);  /* requires SAS 9.4 M4 or later */

output;

end;

run;

Creating Integers for Categorical Variables

Option 4—RAND(“Integer”)

